

LES FONCTIONS MATHÉMATIQUES

SOMME

Définition

La fonction **SOMME** permet **d'additionner** des **nombre**s.

Raccourci : [Ctrl] + [=] (Windows) / [⌘] + [=] (Mac).

Syntaxe

SOMME(Nombre1;[Nombre2];...)

Argument	Description
Nombre1 (obligatoire)	Nombres que tu veux additionner . Ces nombres peuvent être un chiffre (5), une cellule (A3), ou une plage de cellules (A1:A8).
[Nombre2] (facultatif)	Tu peux spécifier jusqu'à 255 nombres de cette façon.

Exemple

Ici, tu veux **calculer** le **total** du **chiffre d'affaires** pour **tous les mois** (cellule **B6**), puis le **cumul** du **CA chaque mois** (**C2 à C5**), et enfin le **total** du CA de **janvier, février et avril** (cellule **C6**).

	A	B	C
1	Mois	Chiffre d'affaires	Cumul du chiffre d'affaires
2	Janvier	1 000 €	1 000 € (2) =SOMME(B\$2:B2)
3	Février	2 000 €	3 000 €
4	Mars	1 500 €	4 500 €
5	Avril	1 500 €	6 000 € =SOMME(B\$2:B5)
6		6 000 € =SOMME(B2:B5) (1) ou =B2+B3+B4+B5	4 500 € (3) =SOMME(B2:B3;B5)
		Total du chiffre d'affaires	Total du chiffre d'affaires de janvier, février et avril

Remarques

- ⇒ (1) La fonction **SOMME** permet de **gagner du temps**, car elle **remplace** =B2+B3+B4+B5.
- ⇒ (2) Si tu veux **calculer** un **cumul**, tu dois **figer uniquement** la **ligne** de la **cellule de référence** B\$2, la cellule suivante B2 étant la **limite** de la **plage**.
- ⇒ (3) Le [:] signifie « **jusqu'à** » et le [;] signifie « **et** » (1000 + 2000 + 1500 = **4500**).

SOMME.SI

Définition

La fonction **SOMME.SI** permet **d'additionner** des **nombres** qui **répondent à un seul critère**.

NB : **SOMME.SI.ENS** (un ou plusieurs critères) est **plus complète** et **plus facile** à utiliser.

Syntaxe

SOMME.SI(Plage;Critère;[Somme_plage])

Argument	Description
Plage (obligatoire)	Plage dans laquelle tu veux rechercher le Critère .
Critère (obligatoire)	Critère que tu veux rechercher dans la Plage. Un critère de texte , de date ou qui contient un symbole mathématique ou logique (*, >, <, =...) doit être placé entre guillemets (""). En revanche, les guillemets ("") ne sont pas nécessaires pour les critères numériques et les références de cellule .
[Somme_plage] (facultatif)	Plage qui contient les nombres que tu veux additionner . Si cet argument est omis , SOMME.SI additionne les cellules de la Plage qui répondent au critère .

Exemple

Ici, tu veux **calculer** le **total des loyers** pour les **biens immobiliers** situés à **Paris** (cellule **B7**) et la **valeur totale** des biens **excédant 350 000 €** (cellule **C7**).

	A	B	C
1	Ville	Loyer	Prix du bien
2	Paris	1 000€	100 000 €
3	Lyon	2 000€	200 000 €
4	Lyon	3 000€	300 000 €
5	Paris	4 000€	400 000 €
6	Paris	5 000€	500 000 €
7		10 000 € =SOMME.SI(A2:A6;"Paris";B2:B6) (1) ou =SOMME.SI(A2:A6;A2;B2:B6)	900 000 € =SOMME.SI(C2:C6;">=350000")
		Total des loyers des biens à Paris	Valeur totale des biens >=350 000 €

Remarque

⇒ (1) Lorsque le **critère** est une **référence de cellule** (ici **A2**), tu dois faire **attention**, car si tu **modifies** le contenu de la cellule **A2**, alors le **critère se met à jour** (si **"Paris"** devient **"Lyon"** dans **A2**, alors le **critère** de la fonction **SOMME.SI** devient **"Lyon"**).

SOMME.SI.ENS

Définition

La fonction **SOMME.SI.ENS** permet **d'additionner** des **nombres** qui **répondent** à **un ou plusieurs critères**.

Syntaxe

SOMME.SI.ENS(Somme_plage;Plage_critères1;Critères1;[Plage_critères2;Critères2];...)

Argument	Description
Somme_plage (obligatoire)	Plage qui contient les nombres que tu veux additionner .
Plage_critères1 (obligatoire)	Plage dans laquelle tu veux rechercher le 1^{er} Critère .
Critères1 (obligatoire)	1^{er} Critère que tu veux rechercher dans la Plage_critères1. Un critère de texte , de date ou qui contient un symbole mathématique ou logique (*, >, <, =...) doit être placé entre guillemets (""). En revanche, les guillemets (") ne sont pas nécessaires pour les critères numériques et les références de cellule .
[Plage_critères2;Critères2] (facultatif)	Tu peux spécifier jusqu'à 127 critères de cette façon.

Exemple

Ici, tu veux **calculer** le **total des ventes** pour les **femmes** qui travaillent à **Lyon** (cellule **D7**).

	A	B	C	D
1	Vendeur	Ville	Sexe	Total des ventes
2	Julien	Paris	H	1 000 €
3	Manon	Lyon	F	500 €
4	Nicolas	Lyon	H	750 €
5	Louise	Lyon	F	1 000 €
6	Élodie	Paris	F	250 €
7				1 500 € (1) =SOMME.SI.ENS(D2:D6;C2:C6;"F";B2:B6;"Lyon")

Total des ventes pour les femmes qui travaillent à Lyon (2 critères)

Remarques

⇒ (1) L'**ordre** des **critères** dans la fonction **SOMME.SI.ENS** ne change **rien**.

⇒ Le **critère "<>"** signifie "**différent de rien**" et le **critère ""** signifie "**rien**".